Lists: Haskell vs. Prolog

Mathematically we might write lists as items separated by commas, enclosed in angle-brackets

$$\sigma_0 = \langle\rangle \quad \sigma_1 = \langle 1\rangle \quad \sigma_2 = \langle 1,2\rangle \quad \sigma_3 = \langle 1,2,3\rangle$$

Haskell

```
s0 = []
s1 = 1:[]   or   [1]
s2 = 1:2:[]   or   [1,2]
s3 = 1:2:3:[] or [1,2,3]
```

Prolog

```
S0 = []
S1 = [1]
S2 = [1,2]
```

`1:2:3:[]` is really `(1:(2:(3:[])))`

Patterns

```
[]   (x:xs)   (x:y:xs)        [] [X|Xs] [X,Y|Xs]
     [x]      [x,y]              [X]    [X,Y]
```

An running example: Expressions

We are going to write functions that manipulate expressions in a variety of ways

```
data Expr
 = Val Float  -- single-precision floating-point number
 | Add Expr Expr
 | Mul Expr Expr
 | Sub Expr Expr
 | Dvd Expr Expr
 deriving Show -- makes it possible to see values (DEMO!)
```

$(10 + 5) * 90$ becomes

```
Mul (Add (Val 10) (Val 5)) (Val 90)
```

$10 + (5 * 90)$ becomes

```
Add (Val 10) (Mul (Val 5) (Val 90))
```

An evaluator

We can write a function to calculate the result of these expressions:

```
eval :: Expr -> Float
eval (Val x) = x
eval (Add x y) = eval x + eval y
eval (Mul x y) = ... -- similar to above
-- similarly for Sub and Dvd
```

```
> eval (Add (Val 10) (Mul (Val 5) (Val 90)))
460.0
```

A simplifier

We can write a function to simplify expressions:

```
simp :: Expr -> Expr
simp (Val x) = (Val x)
-- we can use pattern matching in let-expressions!
simp (Add e1 e2) = let (Val x) = simp e1
                       (Val y) = simp e2
                   in Val (x+y)
simp (Dvd x y) = ... -- similar to above
-- similar for Mul and Sub
```

```
> simp (Add (Val 10) (Mul (Val 5) (Val 90)))
Val 460.0
```

Adding Variables to Expressions

Now let's extend our expression datatype to include variables.
First we extend the expression language:

```
data Expr = Val Float
          | Add Expr Expr
          | Mul Expr Expr
          | Sub Expr Expr
          | Dvd Expr Expr
          | Var Id
     deriving Show

type Id = String
```

Simplification again

```
simp (Add e1 e2)
 = let e1' = simp e1
       e2' = simp e2
   in case (e1',e2') of
     (Val 0.0,e)  ->  e
     (e,Val 0.0)  ->  e
     _            ->  Add e1' e2'
simp (Mul e1 e2) = ... -- similar (and Sub,Dvd)
simp e = e -- catches all remaining cases (Val, Var)
```

Evaluating `Expr`s with Variables

Remember our extended expression language:

```
type Id = String
data Expr = Val Float
          | Add Expr Expr
          | Mul Expr Expr
          | Sub Expr Expr
          | Dvd Expr Expr
          | Var Id
     deriving Show
```

We can't fully evaluate these without some way of knowing what values any of the variables (`Var`) have.
We can imagine that `eval` should have a signature like this:

```
eval :: Dictionary Id Float -> Expr -> Float
```

It now has a new (first) argument, a `Dictionary` that associates `Float` (datum values) with `Id` (key values).

How to model a lookup dictionary?

A *dictionary* maps keys to datum values

- An obvious approach is to use a list of key/datum pairs:

  ```
  type Dictionary k d = [ (k, d) ]
  ```

- Defining a link between key and datum is simply cons-ing such a pair onto the start of the list.

  ```
  define :: Dictionary k d -> k -> d -> Dictionary k d
  define d s v = (s,v):d
  ```

- Lookup simply searches along the list:

  ```
  find :: Eq k => Dictionary k d -> k -> Maybe d
  find [] _ = Nothing
  find ( (s,v) : ds ) name | name == s = Just v
                           | otherwise = find ds name
  ```

  We need to handle the case when the key is not present. This is the role of the `Maybe` type.

Maybe (Prelude)

```
data Maybe a
  =  Nothing
  |  Just a
  deriving (Eq, Ord, Read, Show)

maybe              :: b -> (a -> b) -> Maybe a -> b
maybe n f Nothing  =  n
maybe n f (Just x) =  f x
```

Maybe (Data.Maybe)

```
isJust                  :: Maybe a -> Bool
isJust (Just a)         =  True
isJust Nothing          =  False

isNothing               :: Maybe a -> Bool
isNothing               =  not . isJust

fromJust                :: Maybe a -> a
fromJust (Just a)       =  a
fromJust Nothing        =  error "Maybe.fromJust: Nothing"

fromMaybe               :: a -> Maybe a -> a
fromMaybe d Nothing     =  d
fromMaybe d (Just a)    =  a
```

Dictionary at work

Building a simple dictionary that maps key `"speed"` to datum `20.0`.

```
> define [] "speed" 20.0
[ ("speed", 20.0) ]
> find (define [] "speed" 20.0) "speed"
Just 20.0
> find [] "speed"
Nothing
```

Extending the evaluator

```
eval :: Dictionary Id Float -> Expr -> Float
eval _ (Val x) = x
eval d (Var i) = fromJust (find d i)
eval d (Add e1 e2) = eval d e1 + eval d e2
-- similar for Add, Mul, Dvd
fromJust (Just a) = a
```

We are back to simpler code (no need for `case ... of ...`)

Expr Pretty-Printing

We can write something to print the expression in a more "friendly" infix style:

```
iprint :: Expr -> String
iprint (Val x) = show x
iprint (Var x) = x
iprint (Dvd x y) = "("++(iprint x)++"/"++iprint y++")"
-- similar for Add, Mul, Sub
```

There are many ways in which this could be made much prettier.